

S E R V I C E N O T E

SUPERSEDES: None

8752C Network Analyzer

Serial Numbers: 0000A00000 / 9999Z99999

8752C Firmware Revision 6.14 requires GPIB program to perform factory calibration (Service Test 53)

Parts Required: None

Situation:

For 8752C network analyzers which have Firmware Revision 5.48 and earlier, factory calibration of the instrument (Service Test #53) can be performed manually (the procedure is detailed in the 8752C Service Guide). However, the new firmware revision 6.14 contains a slight glitch in the algorithm for this particular test such that it must be programmed via the GPIB bus to ensure proper calibration.

Solution / Action:

The HP BASIC program that is linked to this document will control the 8752C to ensure proper factory calibration with Service Test #53. This program will run on a 9000 Series 300 or Series 700 workstation and is also compatible with the PC-based HP BASIC for Windows. Clicking on this link will display the contents of the program in your web browser; you can then click the "File" menu on your browser and select "Save As" to save the program to your floppy or hard disk as an ASCII text file.

Continued

DATE: August 1998

ADMINISTRATIVE INFORMATION

| | | |
|------------------------------|---------|-------------------------|
| SERVICE NOTE CLASSIFICATION: | | |
| INFORMATION ONLY | | |
| AUTHOR: | ENTITY: | ADDITIONAL INFORMATION: |
| BH | 5320 | |



```
100  ASSIGN @Vna TO 716
200  INTEGER Stat
210  DISP "Presetting the analyzer...please wait"
220  OUTPUT @Vna;"OPC?;PRES;"
230  ENTER @Vna;Stat
250  INPUT "Enter 50 or 75 for the test set impedance of your 8752C",Z0
251  OUTPUT @Vna;"TEST53;"
252  OUTPUT @Vna;"EXET;"
253  WAIT 1
254  OUTPUT @Vna;"TESR2;"
255  DISP "Clearing all registers...please wait"
257  WAIT 6
258  IF Z0=75 THEN
259  DISP "Connect the Open (including center pin) to the REFL Port, then press F2 to
      continue"
260  ELSE
261  DISP "Connect the Open to the REFL Port, then press F2 to continue."
262  END IF
263  PAUSE
264  DISP "MEASURING OPEN"
265  OUTPUT @Vna;"CLASS11A;OPC?;STANB;"
266  ENTER @Vna;Stat
267  OUTPUT @Vna;"OPC?;DONE;"
268  ENTER @Vna;Stat
272  DISP "Connect the Short to the REFL Port, then press F2."
273  PAUSE
275  DISP "MEASURING SHORT"
276  OUTPUT @Vna;"CLASS11B;OPC?;STANB;"
277  ENTER @Vna;Stat
279  OUTPUT @Vna;"OPC?;DONE;"
280  ENTER @Vna;Stat
284  DISP "Connect the Load to the REFL Port, then press F2."
285  PAUSE
287  DISP "MEASURING LOAD"
288  OUTPUT @Vna;"OPC?;CLASS11C;"
289  ENTER @Vna;Stat
291  DISP "COMPUTING REFLECTION CAL COEFFICIENTS"
292  OUTPUT @Vna;"SAV1;"
293  WAIT 5
296  DISP "Connect the Thru from the REFL Port to the TRANS Port, then press F2."
297  PAUSE
299  DISP "MEASURING THRU"
300  OUTPUT @Vna;"OPC?;STANE;"
301  ENTER @Vna;Stat
302  DISP "COMPUTING TRANSMISSION CAL COEFFICIENTS"
303  OUTPUT @Vna;"OPC?;DONE;"
304  ENTER @Vna;Stat
305  DISP "DONE"
307  CLEAR @Vna
315  LOCAL @Vna
317  END
```